
1. [bookmark: _Toc122434485][bookmark: _Toc96339548]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of IMS test case 7.23 which is part of the iwd-TTCN3-B2020-09_D21wk49.zip test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact: 	Parikshit Bhise
	Parikshit.bhise@rohde-schwarz.com		
	
0. [bookmark: _Toc93474922][bookmark: _Toc93571695][bookmark: _Toc94512040][bookmark: _Toc94621776][bookmark: _Toc96339549]Verification Test Summary
Test Case:	7.23.NR5GC
ATS Version:	iwd-TTCN3-B2020-09_D21wk49.zip
System Simulator used:	Rohde & Schwarz 5G Protocol Conformance Test platform
UE used:	Hisilicon Balong 5000 UE, Samsung SM-G996U UE
Verification Status:	PASS


1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc96339550]Corrections required

1. [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc532983738]Correction to function f_IMS_MOCallSetup_Video_Step2
	Function name
	f_IMS_MOCallSetup_Video_Step2

	Reason for change
	Addition of video is performed with a re-INVITE hence condition A11 is not met.

	Summary of change
	Set condition A11 to false.


	TTCN module
	IMS_Procedures_Video.ttcn

	MCC160 Comment
	Accepted



Before 
<<SKIPPED CODE>>
    select (p_VideoCallType) {
      case (SetupVideoCall) {
        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAudioVideo_RX(p_AudioCodecList, p_VideoCodecList, p_WithPreconditions);
        v_Context := A_2_1_A4;
      }
      case (AddVideo) {
        v_AudioCodec := p_AudioCodecList[0];                /* @sic R5-213073 sic@ */
        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAddVideo_RX(v_AudioCodec, p_VideoCodecList, p_WithPreconditions);
        v_Context := A_2_1_A5;
      }
    }
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
    select (p_VideoCallType) {
      case (SetupVideoCall) {
        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAudioVideo_RX(p_AudioCodecList, p_VideoCodecList, p_WithPreconditions);
        v_Context := A_2_1_A4;
      }
      case (AddVideo) {
        v_A11 := false;  //WA#WI=918780 Change 1: Add video is a re-INVITE hence condition A11 is not met. v_A11 is used to add a check to AcceptContact.
        v_AudioCodec := p_AudioCodecList[0];                /* @sic R5-213073 sic@ */
        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAddVideo_RX(v_AudioCodec, p_VideoCodecList, p_WithPreconditions);
        v_Context := A_2_1_A5;
      }
    }
<<SKIPPED CODE>>

1. Correction to function f_IMS_MTCallSetup_Video_SendInvite
	Function name
	f_IMS_MTCallSetup_Video_SendInvite

	Reason for change
	With current implementation INVITE will only be setup with preconditions.

	Summary of change
	Added input paramterer to f_IMS_InviteRequest_MessageHeaderTX to include preconditinns or not depending of test case needs.


	TTCN module
	IMS_Procedures_Video.ttcn

	MCC160 Comment
	Accepted



Before 
<<SKIPPED CODE>>
    select (p_VideoCallType) {
      case (SetupVideoCall) { v_Context := A_2_1_A4; } /* @sic R5-170782; R5-172930: SetupVideoCall sic@ */
      case else             { v_Context := A_2_1_A5; }
    }
    if (isvalue(p_ContactUrl)) {
      v_ContactUrl := valueof(p_ContactUrl);
    } else {
      v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();
    }
    
    if (not f_SDP_MediaDescr_CheckWithPreconditions(p_SdpOffer.media_list[0])) {       /* @sic R5-216215: no precondition tag in case of A.16.2 sic@ */
      v_OptionTagsForSupported := v_OptionTagsForSupported & tsc_OptionTagList_precondition;
    }

    v_MessageHeader_Invite  := f_IMS_InviteRequest_MessageHeaderTX(-, v_Context);                        /* @sic R5-176209: p_SrvccAlertingSupported := pc_IMS_SRVCCAlert for video too (i.e. v_SrvccAlerting removed) sic@ */
    v_InviteRequest := cs_INVITE_Request(v_ContactUrl, v_MessageHeader_Invite, cs_MessageBody_SDP(p_SdpOffer));
    v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMT(v_InviteRequest, p_SdpOffer);
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
    select (p_VideoCallType) {
      case (SetupVideoCall) { v_Context := A_2_1_A4; } /* @sic R5-170782; R5-172930: SetupVideoCall sic@ */
      case else             { v_Context := A_2_1_A5; }
    }
    if (isvalue(p_ContactUrl)) {
      v_ContactUrl := valueof(p_ContactUrl);
    } else {
      v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();
    }
    
    if (f_SDP_MediaDescr_CheckWithPreconditions(p_SdpOffer.media_list[0])) {       /* @sic R5-216215: no precondition tag in case of A.16.2 sic@ */ //WA#WI=918780 Change 2: Preconditions tags are always included indenpendant previous SDP message.
      v_OptionTagsForSupported := v_OptionTagsForSupported & tsc_OptionTagList_precondition;
    }

    v_MessageHeader_Invite  := f_IMS_InviteRequest_MessageHeaderTX(v_OptionTagsForSupported, v_Context);                        /* @sic R5-176209: p_SrvccAlertingSupported := pc_IMS_SRVCCAlert for video too (i.e. v_SrvccAlerting removed) sic@ */
    v_InviteRequest := cs_INVITE_Request(v_ContactUrl, v_MessageHeader_Invite, cs_MessageBody_SDP(p_SdpOffer));
    v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMT(v_InviteRequest, p_SdpOffer);
<<SKIPPED CODE>>

1. Correction to function f_BuildDefaultQoSRules
	Function name
	f_BuildDefaultQoSRules

	Reason for change
	According to TS 38.508-1 Table 4.8.2.1-7: Reference QoS rule #7, Rule operationcode should be ‘001’B.


	Summary of change
	
Change Rule operation code to value specified in TS 38.508-1 Table 4.8.2.1-7.

	TTCN module
	NG_NAS_Templates.ttcn

	MCC160 Comment
	Accepted



Before 
<<SKIPPED CODE>>
  function f_BuildDefaultQoSRules (charstring p_RuleNum,
                                   template (omit) IEI8_Type p_IEI := omit) return template (value) QoS_Rules
  {
    var template (value) QoS_Rules v_QoS_Rules;
    var template (value) QoS_Rule v_VoiceQoS := cs_QoS_Rule('03'O, int2bit(7, 6), '01'O, -, cs_PacketFilterList_MediaUDP('0110'B), '0'B); // @sic R5-210627 sic@;
    
    if (p_RuleNum == "7_Voice") {
      v_QoS_Rules := cs_QoS_Rules ({v_VoiceQoS}, p_IEI) // @sic R5-213440 sic@
    } else if (p_RuleNum == "7_Video") {
      v_QoS_Rules := cs_QoS_Rules ({cs_QoS_Rule('04'O, int2bit(8, 6), '02'O, '011'B, cs_PacketFilterList_MediaUDP('0111'B), '0'B)}, p_IEI) // @sic R5-217796 sic@
    } else if (p_RuleNum == "7_VoiceVideo") {
      v_QoS_Rules := cs_QoS_Rules ({v_VoiceQoS, cs_QoS_Rule('04'O, int2bit(8, 6), '02'O, -, cs_PacketFilterList_MediaUDP_VideoCall('0111'B), '0'B)}, p_IEI) // @sic R5-210627, R5-216161 sic@
    } else {
      v_QoS_Rules := cs_QoS_Rules ({f_BuildDefaultQoSRule(p_RuleNum)}, p_IEI);
    }
    return v_QoS_Rules;
  }
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_BuildDefaultQoSRules (charstring p_RuleNum,
                                   template (omit) IEI8_Type p_IEI := omit) return template (value) QoS_Rules
  {
    var template (value) QoS_Rules v_QoS_Rules;
    var template (value) QoS_Rule v_VoiceQoS := cs_QoS_Rule('03'O, int2bit(7, 6), '01'O, -, cs_PacketFilterList_MediaUDP('0110'B), '0'B); // @sic R5-210627 sic@;
    
    if (p_RuleNum == "7_Voice") {
      v_QoS_Rules := cs_QoS_Rules ({v_VoiceQoS}, p_IEI) // @sic R5-213440 sic@
    } else if (p_RuleNum == "7_Video") {//WA#WI=918780 Change 3: Rule operation code should be '001'B.
      v_QoS_Rules := cs_QoS_Rules ({cs_QoS_Rule('04'O, int2bit(8, 6), '02'O, -, cs_PacketFilterList_MediaUDP('0111'B), '0'B)}, p_IEI) // @sic R5-217796 sic@
    } else if (p_RuleNum == "7_VoiceVideo") {
      v_QoS_Rules := cs_QoS_Rules ({v_VoiceQoS, cs_QoS_Rule('04'O, int2bit(8, 6), '02'O, -, cs_PacketFilterList_MediaUDP_VideoCall('0111'B), '0'B)}, p_IEI) // @sic R5-210627, R5-216161 sic@
    } else {
      v_QoS_Rules := cs_QoS_Rules ({f_BuildDefaultQoSRule(p_RuleNum)}, p_IEI);
    }
    return v_QoS_Rules;
  }
<<SKIPPED CODE>>

1. Correction to function f_NR5GC_ModifyPDUSession_AddVideoToSpeech
	Function name
	f_NR5GC_ModifyPDUSession_AddVideoToSpeech

	Reason for change
	Test case requires two DRBs, for audio and video. Current number of elements added to list to assign DRB ID for video is one. This will provoke a runtime error as script is trying to access a position with no element.

	Summary of change
	
Created a DRB List with 2 elements. One for audio and another one for video.

	TTCN module
	NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	Only one DRB is added.  Alternative solution implemented: accessed the first element in the list rather than the second: v_NewDRBIdVideo := v_DRBList[0];



Before 
<<SKIPPED CODE>>
    v_DRBList := f_NR_GetNextDRBIds(v_GSM_MobilityInfo, false, 1);
    v_NewDRBIdVideo := v_DRBList[1];
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
    v_DRBList := f_NR_GetNextDRBIds(v_GSM_MobilityInfo, false, 2);
    v_NewDRBIdVideo := v_DRBList[1];
<<SKIPPED CODE>>

1. Correction to function f_IMS_InviteResponse_183_MessageHeaderTX
	Function name
	f_IMS_InviteResponse_183_MessageHeaderTX

	Reason for change
	Change 1: It may happen that HostPort is included by UE in INVITE. In that case SS should include it also in 183 message Contact header.
Change 2: Test case has the singularity that being a MT, UE sends the reINVITE to add video media to the call. This will cause RouteBody to be wrongly set, Dialog is MT but adding video call flow is UE initiated.  

	Summary of change
	Change 1: Added input parameter to add HostPort to Contact header.
Change 2:Added an statement to invert  RouteBody in case that initial dialog is of MT nature. 


	TTCN module
	IMS_NR5GC_CallControlTestcases.ttcn

	MCC160 Comment
	Change 1: 
Accepted in principle even though the argumentation is not correct:
As for change 2 it is that the call establishment has been MT. Therefore the UE comes up with the request URI of the re-INVITE at step 2 being the same URI as provided in the Contact header field of the MT INVITE creating the dialog.
In the responses following to the re-INVITE the SS shall not change this URI  in the 183 response and 200 response at step 7 the URI in Contact header field shall be the same as in the INVITE creating the dialog.
As additional change the request URI of the re-INVITE needs to be checked against the contact URI of the INVITE creating the dialog.
Change 2: 
Accepted in principle; implemented using the new function f_IMS_Response_RouteSet_RouteBodyList_TX which is also applied in f_IMS_OtherResponse_200_MessageHeaderTX.
Necessary prose changes:
· 7.23 step 2: INVITE
Request URI shall be the same as provided by the SS in the Contact header field of the MT INVITE creating the dialog

· 7.23 step 4: 183 Session Progress, 7.23 step 7: 200 OK for INVITE
- Contact header field shall be the same as used in the INVITE creating the dialog
- Record-Route: RouteBody shall be as proposed in 2.5 change 2




Before 
<<SKIPPED CODE>>
  function f_IMS_InviteResponse_183_MessageHeaderTX(INVITE_Request p_InviteRequest,
                                                    charstring p_ContactUri := px_IMS_CalleeContactUri,
                                                    template (omit) SemicolonParam_List p_AditionalContactParams := omit,
                                                    OptionTag_List p_OptionTagsForRequire := tsc_OptionTagList_100rel_precondition,
                                                    IMS_CallType_Type p_CallType := NORMAL_CALL,
                                                    boolean p_A6 := pc_IMS_Video_FeatureTag,
                                                    template (omit) ContentType p_ContentType := cs_ContentTypeSDP) runs on IMS_PTC return template (value) MessageHeader
  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5s150180: tsc_IMS_VideoFeature replaced by pc_IMS_Video_FeatureTag sic@ */
    /* @sic R5s150692 change 7: parameter p_CallType sic@ */
    /* @sic R5-174544: SessionId removed sic@ */
    /* @sic R5s190882: p_OptionTagsForRequire := tsc_OptionTagList_100rel_precondition instead of p_AdditionalOptionTagsForRequire := tsc_OptionTagList_Empty; p_Earlymedia removed; p_ContentType added  sic@ */

    var MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;
    var template (value) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderTX(v_MessageHeader_Invite);  /* @sic R5s190882: f_IMS_BaseResponse_MessageHeaderTX sic@ */
    var template (omit) SemicolonParam_List v_ContactParams := p_AditionalContactParams;
    var template (value) RouteBody_List v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType);     /* @sic R5s150692 change 7 sic@ */

    v_MessageHeader_Response.recordRoute     := cs_RecordRoute(v_RouteBodyList);
    v_MessageHeader_Response.contentType     := p_ContentType;

    if (lengthof(p_OptionTagsForRequire) > 0) {                                              /* @sic R5s190882 sic@ */
      v_MessageHeader_Response.require       := cs_Require(p_OptionTagsForRequire);          // @sic R5s130333 change 3 sic@  @sic R5s190882 sic@
      if (f_MessageHeader_CheckRequire(valueof(v_MessageHeader_Response), c_tag100rel)) {    /* -> reliable provisional responses acc. RFC 3262; condition A3 @sic R5s190882 sic@ */
        v_MessageHeader_Response.rSeq        := cs_RSeq(tsc_IMS_RSeqNumFor183);
      }
    }

    if (p_CallType == ASSERTED_EMERGENCY) {    // Condition A5 @sic R5s150692 change 7 sic@
      v_MessageHeader_Response.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(tsc_IMS_Emergency_TelUri));
    }
    if (pc_BSRVCC) {  // @sic R5s140962 sic@
      v_MessageHeader_Response.featureCaps   := cs_FeatureCaps({ cs_GenericParam(tsc_IMS_FeatureCap_SrvccPreAlerting)});  // @sic R5-145796 sic@
    }
    if (p_A6) { // @sic R5s130109 sic@
      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("video"));
    }
    v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("audio"));                           // @sic R5-165943 sic@
    v_MessageHeader_Response.contact         := cs_Contact(f_SIP_BuildSipUri_TX(p_ContactUri), v_ContactParams);

    return v_MessageHeader_Response;
  }
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_IMS_InviteResponse_183_MessageHeaderTX(INVITE_Request p_InviteRequest,
                                                    charstring p_ContactUri := px_IMS_CalleeContactUri,
                                                    template (omit) SemicolonParam_List p_AditionalContactParams := omit,
                                                    OptionTag_List p_OptionTagsForRequire := tsc_OptionTagList_100rel_precondition,
                                                    IMS_CallType_Type p_CallType := NORMAL_CALL,
                                                    boolean p_A6 := pc_IMS_Video_FeatureTag,
                                                    template (omit) ContentType p_ContentType := cs_ContentTypeSDP) runs on IMS_PTC return template (value) MessageHeader
  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5s150180: tsc_IMS_VideoFeature replaced by pc_IMS_Video_FeatureTag sic@ */
    /* @sic R5s150692 change 7: parameter p_CallType sic@ */
    /* @sic R5-174544: SessionId removed sic@ */
    /* @sic R5s190882: p_OptionTagsForRequire := tsc_OptionTagList_100rel_precondition instead of p_AdditionalOptionTagsForRequire := tsc_OptionTagList_Empty; p_Earlymedia removed; p_ContentType added  sic@ */

    var MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;
    var template (value) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderTX(v_MessageHeader_Invite);  /* @sic R5s190882: f_IMS_BaseResponse_MessageHeaderTX sic@ */
    var template (omit) SemicolonParam_List v_ContactParams := p_AditionalContactParams;
    var template (value) RouteBody_List v_RouteBodyList;     /* @sic R5s150692 change 7 sic@ */
    var template (omit) integer v_HostPort := omit; 

      if (f_IMS_PTC_ImsInfo_DialogIsMO()) {      
        v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType);    
      } else {
        v_RouteBodyList := f_IMS_RouteSet_ReverseOrder(f_IMS_RouteSet_MT_Call_TX());
      }
    
    if (isvalue (p_InviteRequest.requestLine.requestUri.components.sip.hostPort.portField)){ 
        v_HostPort := p_InviteRequest.requestLine.requestUri.components.sip.hostPort.portField;
    }

    v_MessageHeader_Response.recordRoute     := cs_RecordRoute(v_RouteBodyList);
    v_MessageHeader_Response.contentType     := p_ContentType;

    if (lengthof(p_OptionTagsForRequire) > 0) {                                              /* @sic R5s190882 sic@ */
      v_MessageHeader_Response.require       := cs_Require(p_OptionTagsForRequire);          // @sic R5s130333 change 3 sic@  @sic R5s190882 sic@
      if (f_MessageHeader_CheckRequire(valueof(v_MessageHeader_Response), c_tag100rel)) {    /* -> reliable provisional responses acc. RFC 3262; condition A3 @sic R5s190882 sic@ */
        v_MessageHeader_Response.rSeq        := cs_RSeq(tsc_IMS_RSeqNumFor183);
      }
    }

    if (p_CallType == ASSERTED_EMERGENCY) {    // Condition A5 @sic R5s150692 change 7 sic@
      v_MessageHeader_Response.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(tsc_IMS_Emergency_TelUri));
    }
    if (pc_BSRVCC) {  // @sic R5s140962 sic@
      v_MessageHeader_Response.featureCaps   := cs_FeatureCaps({ cs_GenericParam(tsc_IMS_FeatureCap_SrvccPreAlerting)});  // @sic R5-145796 sic@
    }
    if (p_A6) { // @sic R5s130109 sic@
      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("video"));
    }
    v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("audio"));                           // @sic R5-165943 sic@
    v_MessageHeader_Response.contact         := cs_Contact(f_SIP_BuildSipUri_TX(p_ContactUri, v_HostPort), v_ContactParams);

    return v_MessageHeader_Response;
  }
<<SKIPPED CODE>>

MCC160 Implementation – Change 1
	  function f_TC_7_23_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MT Voice Call / add video and remove video / without preconditions at both originating UE and terminating UE / 5GS
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;
    var charstring v_CallIdentification;
    var charstring v_ContactUrlStrSS;
    var SipUrl v_ContactSipUrlSS;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);

    f_IMS_CC_StartCall(IPCAN_MT_SpeechCall);

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_NR5GC_A_5_X_Step1(tsc_SDP_WithoutPreconditions);
    v_ContactSipUrlSS := f_MessageHeader_GetContactSipUrl(v_InviteRequestWithSdp.Invite.msgHeader);
    v_ContactUrlStrSS := f_SIP_SipUrl2Str(v_ContactSipUrlSS);
    f_IMS_MTCallSetup_NR5GC_A_5_2_Step2_10(v_InviteRequestWithSdp);
    v_CallIdentification := f_UT_GetCallId(MMI);

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@
    f_UT_AddVideoToIMSCall(MMI, v_CallIdentification);

    // @siclog "Step 2" siclog@                 INVITE
    v_InviteRequestWithSdp := f_IMS_MOCallSetup_Video_Step2(AddVideo, -, v_ContactSipUrlSS, tsc_IMS_SDP_AudioCodecList_EVS_only, tsc_IMS_SDP_VideoCodecList_H265_H264, tsc_SDP_WithoutPreconditions);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // @siclog "Step 3 - 4" siclog@             100 Trying, 183 Session Progress
    v_ResponseWithSdp := f_IMS_MOCallSetup_Video_Step3_4(AddVideo, v_InviteRequestWithSdp, EVS_16000, H265_90000, v_ContactUrlStrSS);

    // @siclog "Step 5 - 6" siclog@             PRACK, 200 OK
    f_IMS_MOCallSetup_ReceivePRACK_Send200OK(v_InviteRequestWithSdp, v_ResponseWithSdp.Response.msgHeader);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    // @siclog "Step 6A" siclog@
    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MO_AddVideo);
    f_IMS_IPCAN_WaitForTrigger(IPCAN);
      
    // @siclog "Step 7 - 8" siclog@             200 OK, ACK
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp, v_ContactUrlStrSS);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    // @siclog "Step 9 - 13" siclog@            Release Video
    f_IMS_VideoCall_NR5GC_ReleaseVideoMT(v_ResponseWithSdp.SdpMessage, tsc_SDP_WithoutPreconditions);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    // @siclog "Step 14 - 16" siclog@           MMI (Release Call), BYE, 200 OK
    f_IMS_CallReleaseMO(v_InviteRequestWithSdp.Invite);
    
    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MT_SpeechCall);
  }



MCC160 Implementation – Change 2
	  function f_IMS_Response_RouteSet_RouteBodyList_TX(IMS_CallType_Type p_CallType := NORMAL_CALL,
                                                    template (omit) IMS_PTC_DialogIndex_Type p_DialogIndex := omit) runs on IMS_PTC return template (value) RouteBody_List
  {
    var template (value) RouteBody_List v_RouteBodyList;

    if (f_IMS_PTC_ImsInfo_DialogIsMO()) {      
      v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType);                 /* @sic R5s150692 change 17 sic@ */
    } else {
      v_RouteBodyList := f_IMS_RouteSet_ReverseOrder(f_IMS_RouteSet_MT_Call_TX());
    }
    return v_RouteBodyList;
  }

	  function f_IMS_InviteResponse_183_MessageHeaderTX(INVITE_Request p_InviteRequest,
                                                    charstring p_ContactUri := px_IMS_CalleeContactUri,
                                                    template (omit) SemicolonParam_List p_AditionalContactParams := omit,
                                                    OptionTag_List p_OptionTagsForRequire := tsc_OptionTagList_100rel_precondition,
                                                    IMS_CallType_Type p_CallType := NORMAL_CALL,
                                                    boolean p_A6 := pc_IMS_Video_FeatureTag,
                                                    template (omit) ContentType p_ContentType := cs_ContentTypeSDP) runs on IMS_PTC return template (value) MessageHeader
  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5s150180: tsc_IMS_VideoFeature replaced by pc_IMS_Video_FeatureTag sic@ */
    /* @sic R5s150692 change 7: parameter p_CallType sic@ */
    /* @sic R5-174544: SessionId removed sic@ */
    /* @sic R5s190882: p_OptionTagsForRequire := tsc_OptionTagList_100rel_precondition instead of p_AdditionalOptionTagsForRequire := tsc_OptionTagList_Empty; p_Earlymedia removed; p_ContentType added  sic@ */

    var MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;
    var template (value) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderTX(v_MessageHeader_Invite);  /* @sic R5s190882: f_IMS_BaseResponse_MessageHeaderTX sic@ */
    var template (omit) SemicolonParam_List v_ContactParams := p_AditionalContactParams;
    var template (value) RouteBody_List v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType);     /* @sic R5s150692 change 7 sic@ */
    var template (value) RouteBody_List v_RouteBodyList := f_IMS_Response_RouteSet_RouteBodyList_TX(p_CallType);

    v_MessageHeader_Response.recordRoute     := cs_RecordRoute(v_RouteBodyList);
    v_MessageHeader_Response.contentType     := p_ContentType;

    if (lengthof(p_OptionTagsForRequire) > 0) {                                              /* @sic R5s190882 sic@ */
      v_MessageHeader_Response.require       := cs_Require(p_OptionTagsForRequire);          // @sic R5s130333 change 3 sic@  @sic R5s190882 sic@
      if (f_MessageHeader_CheckRequire(valueof(v_MessageHeader_Response), c_tag100rel)) {    /* -> reliable provisional responses acc. RFC 3262; condition A3 @sic R5s190882 sic@ */
        v_MessageHeader_Response.rSeq        := cs_RSeq(tsc_IMS_RSeqNumFor183);
      }
    }

    if (p_CallType == ASSERTED_EMERGENCY) {    // Condition A5 @sic R5s150692 change 7 sic@
      v_MessageHeader_Response.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(tsc_IMS_Emergency_TelUri));
    }
    if (pc_BSRVCC) {  // @sic R5s140962 sic@
      v_MessageHeader_Response.featureCaps   := cs_FeatureCaps({ cs_GenericParam(tsc_IMS_FeatureCap_SrvccPreAlerting)});  // @sic R5-145796 sic@
    }
    if (p_A6) { // @sic R5s130109 sic@
      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("video"));
    }
    v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("audio"));                           // @sic R5-165943 sic@
    v_MessageHeader_Response.contact         := cs_Contact(f_SIP_BuildSipUri_TX(p_ContactUri), v_ContactParams);

    return v_MessageHeader_Response;
  }

	  function f_IMS_OtherResponse_200_MessageHeaderTX(MessageHeader p_MessageHeader_Request,
                                                   template (omit) ContentType p_ContentType := omit,
                                                   template (omit) charstring p_EmergencyTelURI := omit,
                                                   charstring p_ContactUri := px_IMS_CalleeContactUri,
                                                   boolean p_PreconditionsRequired := false,
                                                   IMS_CallType_Type p_CallType := NORMAL_CALL)
    runs on IMS_PTC return template (value) MessageHeader
  { /* @sic R5-130683: common handling for INVITE and UPDATE sic@ */
    /* p_EmergencyTelURI .. shall be set e.g. to tsc_IMS_Emergency_TelUri for condition A6 in A.3.1 */
    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5s150237: parameter p_AdditionalOptionTagsForRequire sic@ */
    /* @sic R5s150692 change 17: parameter p_CallType  sic@ */
    /* @sic R5s150851: no consistency check for p_CallType and p_EmergencyTelURI sic@ */
    /* @sic R5-160794: p_AdditionalOptionTagsForRequire replaced by p_PreconditionIndication sic@ */
    /* @sic R5-174544: SessionId removed sic@ */
    /* @sic R5s210615: p_PreconditionsRequired (boolean) instead of p_PreconditionIndication (IMS_PreconditionIndication_Type) sic@ */
    var template (value) MessageHeader v_MessageHeader_Response := f_IMS_BaseResponse_MessageHeaderTX(p_MessageHeader_Request);  /* @sic R5s190882: f_IMS_BaseResponse_MessageHeaderTX sic@ */
    var charstring v_Method := p_MessageHeader_Request.cSeq.method;
    var boolean v_A1A3 := false;
    var boolean v_A18 := pc_IMS_Video_FeatureTag;
    var boolean v_A19 := false;
    var template (value) RouteBody_List v_RouteBodyList;
    var template (omit) Require v_Require := omit;
    var template (omit) SemicolonParam_List v_ContactParams := omit;

    if (p_PreconditionsRequired and ispresent(p_ContentType)) {    /* @sic R5-160794 sic@ NOTE: responses use Require header even for session modifications */
      v_Require := cs_Require(tsc_OptionTagList_precondition);
    }
    
    select (v_Method) {
      case ("INVITE") {
        v_A1A3 := true;
        v_A19 := true;
      }
      case ("UPDATE") {
        v_A1A3 := true;
      }
    }
    v_MessageHeader_Response.require         := v_Require;                                          /* @sic R5-160794 sic@ */
    v_MessageHeader_Response.contentType     := p_ContentType;

    if (v_A1A3) {
      v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType);             /* @sic R5s150692 change 17 sic@ */
      v_RouteBodyList := f_IMS_Response_RouteSet_RouteBodyList_TX(p_CallType);
      v_MessageHeader_Response.recordRoute   := cs_RecordRoute(v_RouteBodyList);
      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("audio"));                     /* @sic R5-165943 sic@ */
      if (v_A18 and v_A19) {
        v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("video"));                   /* @sic R5-191117 sic@ */
      }
      v_MessageHeader_Response.contact       := cs_Contact(f_SIP_BuildSipUri_TX(p_ContactUri), v_ContactParams);
    }
    if (ispresent(p_EmergencyTelURI)) {    // Condition A6 @sic R5s150851: in case of A6 p_EmergencyTelURI shall be handed over from the calling side; p_CallType is not a criteria sic@
      v_MessageHeader_Response.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(valueof(p_EmergencyTelURI)));
    }
    return v_MessageHeader_Response;
  }




1. Correction to function f_IMS_MOCallSetup_NR5GC_Video_Step6_7
	Function name
	f_IMS_MOCallSetup_NR5GC_Video_Step6_7

	Reason for change
	f_IMS_MOCallSetup_NR5GC_Video_Step6_7 2nd input parameter p_contacUrl is a parameter passed through few functions to finally be used to macht received messages hence nature should be of reception not transmission.

	Summary of change
	
Changed assignment to p_ContactURL to the output of f_SIP_BuildSipUri_RX

	TTCN module
	IMS_Procedures_Video.ttcn

	MCC160 Comment
	Rejected:
f_IMS_MOCallSetup_NR5GC_Video_Step6_7 is not used by this test case 



Before 
<<SKIPPED CODE>>
  function f_IMS_MOCallSetup_NR5GC_Video_Step6_7(VideoCallTypeMO_Type p_VideoCallType,
                                                 IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                 template (value) SipUrl p_ContactUrl := f_SIP_BuildSipUri_TX(px_IMS_CalleeContactUri),
                                                 IMS_CallType_Type p_CallType := NORMAL_CALL) runs on IMS_PTC return SDP_Message
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_IMS_MOCallSetup_NR5GC_Video_Step6_7(VideoCallTypeMO_Type p_VideoCallType,
                                                 IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                 template (value) SipUrl p_ContactUrl := f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri),
                                                 IMS_CallType_Type p_CallType := NORMAL_CALL) runs on IMS_PTC return SDP_Message
<<SKIPPED CODE>>

1. Correction to function f_SIP_BuildSipUri_RX
	Function name
	f_SIP_BuildSipUri_RX

	Reason for change
	Being definition of   Hostport:

             type record HostPort {
                    charstring host optional, // hostname, IPv4 or IPv6
                    integer portField optional // represented as an integer
                }
Input parameter should reflect that and be assigned with “*”.

	Summary of change
	
Assigned “*” to p_Port later assignet to portField to admit any or none value sent by the UE.

	TTCN module
	IMS_SIP_Templates.ttcn

	MCC160 Comment
	Rejected: In cases where it is allowed that the UE comes upt with or without portnumber this can easily be expressed by handing over '*' (see e.g. implementation of f_IMS_INVITE_CalleeUriRX). In addition the proposed change would affect more than 60 places where f_SIP_BuildSipUri_RX is used.



Before 
<<SKIPPED CODE>>
  function f_SIP_BuildSipUri_RX(charstring p_URI,
                                template integer p_Port := omit,
                                template SemicolonParam_List p_UrlParameters := *,
                                template AmpersandParam_List p_Headers := omit) return template (present) SipUrl
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_SIP_BuildSipUri_RX(charstring p_URI,
                                template integer p_Port := *,
                                template SemicolonParam_List p_UrlParameters := *,
                                template AmpersandParam_List p_Headers := omit) return template (present) SipUrl
<<SKIPPED CODE>>

1. Correction to function fl_NR5GC_ModifyPDUSession_RemoveIMS_Media
	Function name
	fl_NR5GC_ModifyPDUSession_RemoveIMS_Media

	Reason for change
	To remove video media QoS rule and flow for video should be deleted. Current implementation modify Rule and delete rule. That can cause UE to reject the PDU session Modification.

It will require also a Prose change for TS 38.508-1 Table 4.9.28.2.3-2

	Summary of change
	
Replaced modfy QoS rule with Delete QoS rule to be sent in PDU Session modification to remove video media from call.

	TTCN module
	NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	Accepted



Before 
<<SKIPPED CODE>>
    // @sic R5-216150 sic@
    if (v_RemoveVoice) { // @sic R5-217798 sic@
      v_QoS_Rules[j] := cs_QoS_DeleteRule('03'O); // @sic R5s210073, R5-213439 sic@
      v_QoS_Flows[j] := cs_QoS_DeleteFlow('000111'B); // @sic R5-213439 sic@
      j := j+1;
    }
    if (v_RemoveVideo) {
      if (v_RemoveVoice) {
        v_QoS_Rules[j] := cs_QoS_DeleteRule('04'O);
      } else {  // @sic R5-217798 sic@ FFS: error in prose
        v_QoS_Rules[j] := cs_QoS_ModifyRule('04'O);
      }
        v_QoS_Flows[j] := cs_QoS_DeleteFlow('001000'B);
    }
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
    // @sic R5-216150 sic@
    if (v_RemoveVoice) { // @sic R5-217798 sic@
      v_QoS_Rules[j] := cs_QoS_DeleteRule('03'O); // @sic R5s210073, R5-213439 sic@
      v_QoS_Flows[j] := cs_QoS_DeleteFlow('000111'B); // @sic R5-213439 sic@
      j := j+1;
    }
    if (v_RemoveVideo) {
<<COMMENTED OUT>>
      /*if (v_RemoveVoice) {
        v_QoS_Rules[j] := cs_QoS_DeleteRule('04'O);
      } else {  // @sic R5-217798 sic@ FFS: error in prose
        v_QoS_Rules[j] := cs_QoS_ModifyRule('04'O);
      }*/
<<COMMENTED OUT>>

        v_QoS_Rules[j] := cs_QoS_DeleteRule('04'O);        
        v_QoS_Flows[j] := cs_QoS_DeleteFlow('001000'B);
    }
<<SKIPPED CODE>>

1. Correction to function f_IMS_VideoCall_NR5GC_ReleaseVideoMT
	Function name
	f_IMS_VideoCall_NR5GC_ReleaseVideoMT

	Reason for change
	Function  f_IMS_VideoCall_NR5GC_ReleaseVideoMT is hardcoded to only support preconditions.

	Summary of change
	
Added input parameters to add the possibility to set preconditions or not depending on test cases needs.

	TTCN module
	IMS_NR5GC_CallControlTestcases.ttcn

	MCC160 Comment
	Accepted



Before 
<<SKIPPED CODE>>
  function f_IMS_VideoCall_NR5GC_ReleaseVideoMT(template (value) SDP_Message p_LatestSdpMessageFromSS) runs on IMS_PTC
  {
    var SipUrl v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);   // The UE's Contact URI
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) SDP_Message v_SdpOffer := p_LatestSdpMessageFromSS;
    var template (value) MessageHeader v_MessageHeaderINVITE;
    var template (present) MessageHeader v_MessageHeader200OK;
    var template (value) MessageHeader v_MessageHeaderACK;

    // Send INVITE
    v_SdpOffer.origin.session_version := f_IMS_PTC_ImsInfo_DialogIncrAndGetSessVersionTX();
    v_SdpOffer.media_list[1].media_field.ports.port_number := 0;
    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(RemoveVideo, v_SdpOffer, v_ContactUrlUE);

    // Receive 100 Trying (optional) and 200 OK
    v_MessageHeaderINVITE := v_InviteRequestWithSdp.Invite.msgHeader;
    v_MessageHeader200OK := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeaderINVITE, cr_ContentTypeSDP);
    f_IMS_ReceiveResponse_OptionalTrying(v_InviteRequestWithSdp.Invite.msgHeader,
                                         v_InviteRequestWithSdp.RoutingInfo.Protocol,
                                         cr_Response(c_statusLine200, v_MessageHeader200OK, cr_MessageBody_SDP));

    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_ReleaseVideo);
    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    // Send ACK
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite);
    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_InviteRequestWithSdp.RoutingInfo.Protocol), cs_ACK_Request(v_ContactUrlUE, v_MessageHeaderACK)));
  }
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_IMS_VideoCall_NR5GC_ReleaseVideoMT(template (value) SDP_Message p_LatestSdpMessageFromSS, boolean p_PreconditionRequiredForSDP := true) runs on IMS_PTC
  {
    var SipUrl v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);   // The UE's Contact URI
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) SDP_Message v_SdpOffer := p_LatestSdpMessageFromSS;
    var template (value) MessageHeader v_MessageHeaderINVITE;
    var template (present) MessageHeader v_MessageHeader200OK;
    var template (value) MessageHeader v_MessageHeaderACK;

    // Send INVITE
    v_SdpOffer.origin.session_version := f_IMS_PTC_ImsInfo_DialogIncrAndGetSessVersionTX();
    v_SdpOffer.media_list[1].media_field.ports.port_number := 0;
    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(RemoveVideo, v_SdpOffer, v_ContactUrlUE);

    // Receive 100 Trying (optional) and 200 OK
    v_MessageHeaderINVITE := v_InviteRequestWithSdp.Invite.msgHeader;
    v_MessageHeader200OK := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeaderINVITE, cr_ContentTypeSDP, p_PreconditionRequiredForSDP); 
    f_IMS_ReceiveResponse_OptionalTrying(v_InviteRequestWithSdp.Invite.msgHeader,
                                         v_InviteRequestWithSdp.RoutingInfo.Protocol,
                                         cr_Response(c_statusLine200, v_MessageHeader200OK, cr_MessageBody_SDP));

    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_ReleaseVideo);
    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    // Send ACK
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite);
    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_InviteRequestWithSdp.RoutingInfo.Protocol), cs_ACK_Request(v_ContactUrlUE, v_MessageHeaderACK)));
  }
<<SKIPPED CODE>>

1. Correction to function f_TC_7_23_NR5GC_IMS1
	Function name
	f_TC_7_23_NR5GC_IMS1

	Reason for change
	Test case is implemented to run with preconditions when should be run without.

	Summary of change
	
Added input parameter to configurestep 9-13 to be run without preconditions.

	TTCN module
	IMS_NR5GC_CallControlTestcases.ttcn

	MCC160 Comment
	Accepted



Before 
<<SKIPPED CODE>>
    // @siclog "Step 7 - 8" siclog@             200 OK, ACK
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    // @siclog "Step 9 - 13" siclog@            Release Video
    f_IMS_VideoCall_NR5GC_ReleaseVideoMT(v_ResponseWithSdp.SdpMessage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    // @siclog "Step 14 - 16" siclog@           MMI (Release Call), BYE, 200 OK
    f_IMS_CallReleaseMO(v_InviteRequestWithSdp.Invite);
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
    // @siclog "Step 7 - 8" siclog@             200 OK, ACK
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    // @siclog "Step 9 - 13" siclog@            Release Video
    f_IMS_VideoCall_NR5GC_ReleaseVideoMT(v_ResponseWithSdp.SdpMessage, tsc_SDP_WithoutPreconditions);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    // @siclog "Step 14 - 16" siclog@           MMI (Release Call), BYE, 200 OK
    f_IMS_CallReleaseMO(v_InviteRequestWithSdp.Invite);
<<SKIPPED CODE>>

1. Correction to function f_NR5GC_IPCAN_AddVideo
	Function name
	f_NR5GC_IPCAN_AddVideo

	Reason for change
	Function f_NR5GC_IPCAN_AddVideo is missing coordination message to inform IMS PTC that video media has been added successfully.

	Summary of change
	
Added coordination message to inform IMS PTC that video media has been added.

	TTCN module
	IPCAN_NR5GC.ttcn

	MCC160 Comment
	Accepted



Before 
<<SKIPPED CODE>>
  function f_NR5GC_IPCAN_AddVideo(IMS_IPCAN_CO_ORD_PORT p_Port,
                                  NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC return TriggerResult_Type

  {
    var TriggerResult_Type v_Result := NORMAL;
    // Steps 2 - 8 the same as for speech call
    f_NR5GC_ModifyPDUSession_AddVideoToSpeech(p_NR_CellId);
    return v_Result;
  }
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_NR5GC_IPCAN_AddVideo(IMS_IPCAN_CO_ORD_PORT p_Port,
                                  NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC return TriggerResult_Type

  {
    var TriggerResult_Type v_Result := NORMAL;
    // Steps 2 - 8 the same as for speech call
    f_NR5GC_ModifyPDUSession_AddVideoToSpeech(p_NR_CellId);
    f_IMS_IPCAN_SendCoOrdMsg(p_Port); 
    return v_Result;
  }
<<SKIPPED CODE>>

1. Correction to function f_NR5GC_IPCAN_ReleaseVideo
	Function name
	f_NR5GC_IPCAN_ReleaseVideo

	Reason for change
	Function f_NR5GC_IPCAN_ReleaseVideo is missing coordination message to inform IMS PTC that video media has been released successfully.

	Summary of change
	
Added coordination message to inform IMS PTC that video media has been released.

	TTCN module
	IPCAN_NR5GC.ttcn

	MCC160 Comment
	Accepted



Before 
<<SKIPPED CODE>>
  function f_NR5GC_IPCAN_ReleaseVideo(IMS_IPCAN_CO_ORD_PORT p_Port,
                                      NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC return TriggerResult_Type

  {
    var TriggerResult_Type v_Result := NORMAL;
    // Steps 2 - 8 the same as for speech call
    f_NR5GC_ModifyPDUSession_RemoveVideoFromSpeech(p_NR_CellId);
    return v_Result;
  }
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_NR5GC_IPCAN_ReleaseVideo(IMS_IPCAN_CO_ORD_PORT p_Port,
                                      NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC return TriggerResult_Type

  {
    var TriggerResult_Type v_Result := NORMAL;
    // Steps 2 - 8 the same as for speech call
    f_NR5GC_ModifyPDUSession_RemoveVideoFromSpeech(p_NR_CellId);
    f_IMS_IPCAN_SendCoOrdMsg(p_Port); 
    return v_Result;
  }
<<SKIPPED CODE>>

1. [bookmark: _Toc93667850][bookmark: _Toc96339551][bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc532983740]Branches executed
The test case was executed on NR band n41 using NR ciphering algorithm nea1 and integrity algorithm nia1 on Rohde & Schwarz 5G Protocol Conformance Test platform with Samsung SM-G996U UE and Hisilicon Balong 5000 UE.
1. [bookmark: _Toc93667851][bookmark: _Toc96339552]Execution Log Files
[bookmark: _Toc88663234][bookmark: _Toc93667852][bookmark: _Toc96339553]HiSilicon
The HiSilicon Balong5000 IMS UE passed this test case on Rohde & Schwarz 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_7.23_Hisilicon.log
In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
[bookmark: _Toc96339554]Samsung SM-G996U
The Samsung SM-G996U UE passed this test case on Rohde & Schwarz 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_7.23_Samsung.log
In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

1. [bookmark: _Toc96339555]References

	R5s220337: Supporting information for addition of IMS 5GS Call Control test case 7.23





